
	What does a list look like?

	 

	
Many programming languages include lists as one of their data types. Lists are an invaluable part of artificial intelligence programming languages and the name of the major AI language, LISP, is a contraction of "LISt Processing". Different programming languages use different notations to express lists. Prolog lists look like: 

     [m42, m40, m25, a21]

Briefly, it consists of items (called "members" or "elements") inside square brackets separated by commas. In this example, each element is an atom, but Prolog doesn't restrict elements to the atom data type. For instance, lists can be lists of lists: 

     [[camel, emu, deer], [tiger, bear, bear]]

or lists of structured objects: 

     [road(m42, [j1, j2, j3, j3a]), road(m40, [j16, j15, j14, j13, j12, j11])]

or a mixture of data types: 

     [road(m42, [j1, j2, j3, j3a, j4, j5]), road(a42, [twycross]), zoo, 

                [marmoset, tamarin, lemur]]

List can even be empty, in which case they look like: 

     []



	

	 

	Why use lists?
 

Lists are an easy way of working with unpredictable situations. They are invaluable in situations where it is difficult, impossible or not worthwhile to predict data to be stored and manipulated. In many artificial intelligence programs, the number of partial solutions to a problem changes dynamically during computation. For instance, near the beginning of a session with a patient, a medical diagnosis system might consider a small number of possible illnesses, which expands to a greater number when more symptoms are discovered, and which finally settles down to one or two possible illnesses. A list would be an ideal data structure for such an application. 
  


	 

	Unification and lists

All the examples of unification seen so far have included only predictable, fixed data structures, such as atomics and structured objects. We can unify one list with another: 

     | ?- [a, b, c] = [Elem1, Elem2, Elem3].

     Elem1 = a,

     Elem2 = b,

     Elem3 = c ?

and because an uninstantiated variable will unify with anything, we can unify a list with a variable: 

     | ?- Var = [a, b, c].

     Var = [a,b,c] ?

As these two examples show, we haven't achieved much. In the first example we had to know exactly how many elements there are in the lists: in the second example we just instantiated Var to the complete list, rather than accessing parts of it. 

In order to be able to use unification with lists of unknown length, we must have an extension to the list notation and so to our way of thinking about lists. 

Rather than thinking of a list as consisting of a certain number of elements, we can think of it as having a head and a tail. At the simplest, the head is the first element of the list and the tail is the remainder. In Prolog, we use the "list constructor" (written as "|") to denote the head and tail. 

     | ?- [a, b, c] = [Head|Tail].

     Head = a,

     Tail = [b,c] ?

The important thing to notice in this example is that the tail of the list is itself a list. In this example, there was one element before the list constructor, but it is possible to have more than one: 

     | ?- [a, b, c] = [Head1, Head2|Tail].

     Tail = [c],

     Head1 = a,

     Head2 = b ?

Again, note that the tail of the list is still a list in its own right, even though it only has one element. What happens if we extend this example to include three elements before the list constructor? 

     | ?- [a, b, c] = [Hd1, Hd2, Hd3|Tail].

     Hd1 = a,

     Hd2 = b,

     Hd3 = c,

     Tail = [] ?

All three elements are unified with variables before the list constructor, so there is nothing with which Tail can be unified. This is shown by Tail being instantiated to the empty list '[]'.

	How are lists processed?
 

We've already stated that the main reason for using lists is their capability of expressing unpredictable and dynamic situations. However, list processing algorithms must be predictable. This may seem paradoxical, but the situation is retrieved by a neat programming trick. 

The most common pattern of list processing divides a list into a head and a tail. Some operation is done on the head, and then the tail is itself divided into a head and a tail and the operation repeated anew. This process can go on until a terminating condition is reached. The most obvious terminating condition is when the list is empty. So, given a list [a, b, c], it would be reduced in four steps: 

1. Head = a, Tail = [b, c] 

2. Head = b, Tail = [c] 

3. Head = c, Tail = [] 

4. List empty ("[]"). 

The processing of this list can be drawn as: 

The shape of this tree suggests that a procedure for working through the list would have the same pattern as a procedure for exploring the network in Modules 3 and 4. That procedure had two rules, of which the first was the terminating condition and the second was the recursive condition. 

The procedure for traversing the network was: 

     % 1

     route(Start, End) :-

          path(Start, End).

     % 2

     route(Start, End) :-

          path(Start, Via),

          route(Via, End).

The first thing we need to do is to write down the form of the head of the rules. The first rule will have the form: 

     % 1 terminating

     dissect_list([]) :-

The second rule will have the form: 

     % 2 recursive

     dissect_list([Head|Tail]) :-

Next we have to write the body of the rules. The body of the first rule will be: 

     write([]), nl.

The body of the second rule will be: 

     write(Head), 

     nl,

     dissect_list(Tail).

Putting this together, we have the procedure: 

     % 1 terminating

     dissect_list([]) :-

          write([]), nl.

     % 2 recursive

     dissect_list([Head|Tail]) :-

          write(Head), 

          nl,

          dissect_list(Tail).

	In the previous section we saw how it was possible to work through a list, taking the head of a list and writing it, then performing the same operation on the tail. The procedure used to implement this included two rules: one recursive and the other the terminating condition. 

Terminating conditions are essential in list processing because they usually represent conditions that we are more interested in. The three patterns that we are interested in are: 

Terminate when the list is empty
We've already seen an example of this in dissect_list/1. The key point is that the computation ends when the empty list ([]) is found. 

Terminate when a specified element is found
Examples of this are searching for a given element in a list, either to discover it is there, or to perform some operation on it, such as replacement. 

Terminate when a specific point is reached
Here, we work through lists until we have reached a specific point (irrespective of what that element is), and then perhaps perform some operation on it. 



Each of the three sections include examples designed to illustrate the topic. However, the examples given aren't necessarily the ones most frequently used by expert programmers, because there are other versions that are more efficient. These are considerations that have to take into account the way that Prolog is implemented on the underlying hardware. In this sense, the previous versions are not "wrong" but less appropriate. 

The topic of efficiency in list processing is taken further in discussion on the topic of accumulators. 

	The three patterns of list processing have been presented. Now we will look again at some of the selected examples. 


	

	app/3 revisited

The definition of app/3 was: 

     % 1 terminating condition

     app([], List, List).

     % 2 recursive

     app([Hd|Tl1], List2, [Hd|Tl3]) :-

          app(Tl1, List2, Tl3).

We could use this to find: 

     | ?- app([a,b,c], [1,2,3], List).

     List=[a,b,c,1,2,3] ? ;

     no

This is the use we would expect of such a procedure: we're asking "what does [a,b,c] and [1,2,3] make when appended?". We can ask an alternative question: what pairs of lists, when appended, make [a,b,c,1,2,3]. 

     | ?- app(List1, List2, [a,b,c,1,2,3]).

     List1 = [] 

     List2 = [a,b,c,1,2,3] ? ;

     List1 = [a] 

     List2 = [b,c,1,2,3] ? ;

     List1 = [a,b] 

     List2 = [c,1,2,3] ? ;

     List1 = [a,b,c] 

     List2 = [1,2,3] ? ;

     List1 = [a,b,c,1] 

     List2 = [2,3] ? ;

     List1 = [a,b,c,1,2] 

     List2 = [3] ? ;

     List1 = [a,b,c,1,2,3] 

     List2 = [] ? ;

     no

We can see that there are seven possible pairs that make up the original list. 

One way of looking at what we have done is that we have used this procedure normally (ie we have found what the appending of two lists will give), and "back-to-front" (ie by finding the question for which know the answer). This way of using procedures in more than one way is one of logic programming's strengths. Given enough imagination the part of the programmer, it is possible to create some very elegant and simple solutions to problems. However, as will be seen later in the course, it is quite easy to get caught out by inadvertently finding infinite branches in the search tree: ie situations where Prolog will recurse forever without finding a solution.   
  



	 

	delete_element/3

The definition of delete_element/3 was given as: 

     % 1 terminating condition

     delete_element(Elem, [Elem|Tail], Tail).

     % 2 recursive

     delete_element(Elem, [Head|Tail1], [Head|Tail2]) :-

          delete_element(Elem, Tail1, Tail2).

We have seen that we can use this to find what a list is with a particular item deleted: 

     | ?- delete_element(b, [a,b,c], [a,c]).

     yes

     | ?- delete_element(c, [a,b,c], List).

     List = [a,b] ? ;

     no

We can also use this procedure to find what lists can be created by adding a single element to it: 

     | ?- delete_element(b, List, [a,c]).

     List = [b,a,c] ? ;

     List = [a,b,c] ? ;

     List = [a,c,b] ? ;

     no

We can also use it to find what pairs of an element and a list can be made from another list: 

     | ?- delete_element(Elem, [a,b,c], List).

     Elem = a 

     List = [b,c] ? ;

     Elem = b 

     List = [a,c] ? ;

     Elem = c 

     List = [a,b] ? ;

     no


  


	 

	Examples that use Prolog arithmetic
 

We have seen two examples of the multiple uses of procedures, but it would be wrong to think that all Prolog procedures can be used in more than one way. Apart from procedures with only one argument and examples which disappear into infinite recursion, another important group of procedures are those that include arithmetic. In Prolog, arithmetic is implemented as an extra-logical extension and is not bi-directional. Simply put, Prolog is happy with an uninstantiated variable on the left-hand side of is/2, but will halt with an instantiation error is there is an uninstantiated variable on the right-hand side. 

In a previous section, we defined memb/3 as: 

     % 1 terminating condition

     memb(Elem, [Elem|_], Cnt) :-

          write(Cnt),

          nl.

     % 2 recursive

     memb(Elem, [_|Tail], Cnt) :-

          Cnt1 is Cnt + 1,

          memb(Elem, Tail, Cnt1).

and described its use to find the position at which a given element occurred. It seems very similar to position/3, which we defined as: 

     % 1 terminating condition

     position(1, Elem, [Elem|_]).

     % 2 recursive

     position(Cnt, Elem, [_|Tail]) :-

          Cnt1 is Cnt - 1,

          position(Cnt1, Elem, Tail).

However, if we use position/3 to perform the functionality of memb/3, or vice versa, we will soon have instantiation errors because the variable Cnt will be uninstantiated. 

The patterns of list processing given so far don't provide a way around this problem. However, there is another pattern of list processing that side-steps the problem and also offers more opportunity to write more efficient recursive programs. Procedures in this pattern are generally called accumulators. 

	This section first looks at how to write accumulators, re-working examples seen in earlier sections, and then examines why accumulators can be more efficient. 



	

	  

	count_elems/2 revisited

We'll start with the simplest procedure that included Prolog arithmetic, although the original version seemed quite satisfactory: 

     % 1 terminating condition

     count_elems([], 0).

     % 2 recursive

     count_elems([Head|Tail], Count) :-

          count_elems(Tail, Sum), 

          Count is Sum + 1.

This uses a variable called Count which acts as a counter of the number of elements in the list. In an accumulator, we add another argument to help the counting. This is the new terminating condition: 

     % 1 terminating condition

     count_elems_acc([], Total, Total).

This simply says that when the list is empty, then the other two arguments are the same. 

The recursive condition is a rearrangement of the old version: 

     % 2 recursive

     count_elems_acc([Head|Tail], Sum, Total) :-

          Count is Sum + 1, 

          count_elems_acc(Tail, Count, Total).

This simply adds one to the counter, while recursing on the Tail of the list and passing the Total on. We can see how the variables are instantiated as recursion deepens by following the output of demo_count_elems_acc/3. Look especially at how Count is incremented and how Total is only instantiated when the terminating condition is reached: 

     | ?- demo_count_elems_acc([a,b,c], 0, Total).

          Before recursion at depth 1

             List is: [a,b,c]

             Count is: 0

             Total is: _425

               Before recursion at depth 2

                  List is: [b,c]

                  Count is: 1

                  Total is: _425

                    Before recursion at depth 3

                       List is: [c]

                       Count is: 2

                       Total is: _425

                         At the termination condition

                            List is: []

                            Count is: 3

                            Total is: 3

                    After recursion at depth 3

                       List is: [c]

                       Count is: 2

                       Total is: 3

               After recursion at depth 2

                  List is: [b,c]

                  Count is: 1

                  Total is: 3

          After recursion at depth 1

             List is: [a,b,c]

             Count is: 0

             Total is: 3

     

     Total = 3 ;

     no

To call the procedure, we can try various versions: 

     | ?- count_elems_acc([a,b,c], 0, Total).

     Total = 3 ? ;

     no

     | ?- count_elems_acc([a,b,c], 0, 4).

     no

     | ?- count_elems_acc([], 0, Total).

     Total = 0 ? ;

     no

     | ?- count_elems_acc(List, 0, 3).

     List = [_191,_197,_203] 

     yes

It is inconvenient to remember to add the literal '0' every time count_elems_acc/3 is called, so we write another rule, count_elems_acc/2, which calls count_elems_acc/3 with the literal: 

     % 1

     count_elems_acc(List, Total) :-

          count_elems_acc(List, 0, Total).

                



	 

	memb/3 and position/3 revisited

We've already commented on the apparent similarity between memb/3 and position/3. In this section, we'll rewrite them in a single, accumulator procedure. 

     % 1 terminating

     pos_memb_acc(Pos, Pos, Elem, [Elem|_]).

     % 2 recursive

     pos_memb_acc(Cnt, Pos, Elem, [_|Tail]) :-

          Cnt1 is Cnt + 1,

          pos_memb_acc(Cnt1, Pos, Elem, Tail).

Like count_elems_acc/3, this uses a counter (called Cnt and Cnt1) and a total (called Pos) to keep track on the position of the element. The recursive condition is true when the counter is incremented and the recursive call is true. The terminating condition is true when the head of the list is unifiable with Elem and the counter and Pos can be unified. 

Again, we'll provide a more convenient calling rule: 

     pos_memb_acc(Pos, Elem, List) :-

          pos_memb_acc(1, Pos, Elem, List).

Try these procedures with the following two queries: 

     | ?- pos_memb_acc(3, Elem, [a,b,c,d,e]).

     | ?- pos_memb_acc(Pos, c, [a,b,c,d,e]). 
 



	 

	Using recursion efficiently

It would be incorrect to think that the only use of accumulators was to allow procedures to be "bi-directional". Recursion is potentially a large user of main memory, because each recursive call requires the state of the computation to be stored and a new state to be computed. In realistically large Prolog programs, recursion is frequently used and so memory consumption could be huge. However, there is a technique of implementing recursion that allows memory to be used very economically. Look at these two versions of a recursive rule: 

     % 2 recursive

     count_elems([Head|Tail], Count) :-

          count_elems(Tail, Sum), 

          Count is Sum + 1.

     % 2 recursive

     count_elems_acc([Head|Tail], Sum, Total) :-

          Count is Sum + 1, 

          count_elems_acc(Tail, Count, Total).

In the first version, the first sub-goal (count_elems(Tail, Sum)) is non-deterministic: that is to say that there could be more than one alternative matching clause in the Prolog program; the second sub-goal is deterministic, because there is only one possible solution to Count is Sum + 1. 

In the second version, the first sub-goal (Count is Sum + 1) is deterministic; the second sub-goal (count_elems(Tail, Sum, Total)) is non-deterministic. 

In Prolog, the efficient use of memory hinges on determinism and non-determinism. Look again at the second version. As the first sub-goal (Count is Sum + 1) is deterministic, when the second, recursive sub-goal (count_elems(Tail, Sum, Total)) is reached, the Prolog interpreter/compiler has no need to "remember" anything about the previous sub-goals. This means that the recursive call can use the memory being used by the current call and so, as recursion deepens, there is no increase in memory used. 

With the first version, it may be that some sub-goal following the recursive sub-goal may be non-deterministic, and so Prolog cannot over-write the memory used for the current recursive call and so requiring the consumption of more main memory. 

This technique of economising on the use of memory during recursion is part of a technique called Tail Recursion Optimisation. Its name refers to the position of the recursive sub-goal at the tail-end of the recursive rule. Experienced Prolog programmers are identifiable by their wide-spread use of tail-recursion. 
  



	  

	reverse/2: an example of efficiency

There is a version of reverse commonly known as "naïve reverse" and often quoted as a benchmark for measuring comparative speeds of Prolog interpreters and compilers. "Naïve reverse" in turn uses append (which we defined as app/3). A definition is: 

     % 1 terminating condition

     reverse_naive([], []).

     % 2 recursive

     reverse_naive([Head|Tail1], Reversed) :-

          reverse_naive(Tail1, Tail2),

          app(Tail2, [Head], Reversed).

This seems a harmless piece of code, but look at the recursive rule. This recursively reverses Tail1 to obtain Tail2 and then appends Head to Tail2. This means that reverse_naive/2 recurses through to the empty list and, at each level of recursion, app/3 also recurses through to the end of the list. If you want an illustration of what this means, simply copy reverse_naive/2 and app/3 into a file and trace a query like: 

     | ?- reverse_naive([a,b,c,d,e,f], Rev).
When I did this, I counted no less than twenty-eight separate calls to procedures, of which, seven were calls to reverse_naive/2 and twenty-one were to app/3. By rewriting reverse_naive/2 as an accumulator, we can drastically reduce the amount of processing: 

     % 1

     reverse_acc(List, Tsil) :-

          reverse_acc(List, [], Tsil).

     % 1 terminating condition

     reverse_acc([], Reversed, Reversed).

     % 2 recursive

     reverse_acc([Head|Tail], Rest, Reversed) :-

          reverse_acc(Tail, [Head|Rest], Reversed).

This version starts the second argument of reverse_acc/3 as an empty list and then uses the list constructor to add Head to the second argument at each stage of recursion. The terminating condition specifies that the second and third arguments should be unified. 

When I traced this version, there were only eight procedure calls. Obviously the accumulator version does far less work and it has the advantage of being tail-recursive, whereas the naïve version isn't, with a corresponding lack of efficiency. 

Beginners at Prolog find it easy to write procedures that aren't tail recursive and don't use accumulators where they should. However, with practise and experience, better habits prevail.

	count_elems_acc/3 revisited

Amongst the queries we gave for count_elems_acc/3 was: 

     | ?- count_elems_acc(List, 0, 3).

     List = [_A,_B,_C] ? 

     yes

Note that we did not ask for any alternative solutions. This was quite deliberate. Look again at the procedures: 

     % 1 terminating condition

     count_elems_acc([], Total, Total).

     % 2 recursive

     count_elems_acc([Head|Tail], Sum, Total) :-

          Count is Sum + 1, 

          count_elems_acc(Tail, Count, Total).

We'll look at the output of the demo version of this program, demo_count_elems_acc/3 for this query: 

     | ?- demo_count_elems_acc(List, 0, 3).

          Before recursion at depth 1

             List is: [_196|_197]

             Count is: 0

             Total is: 3

               Before recursion at depth 2

                  List is: [_708|_709]

                  Count is: 1

                  Total is: 3

                    Before recursion at depth 3

                       List is: [_1480|_1481]

                       Count is: 2

                       Total is: 3

                         At the termination condition

                            List is: []

                            Count is: 3

                            Total is: 3

                    After recursion at depth 3

                       List is: [_1480]

                       Count is: 2

                       Total is: 3

               After recursion at depth 2

                  List is: [_708,_1480]

                  Count is: 1

                  Total is: 3

          After recursion at depth 1

             List is: [_196,_708,_1480]

             Count is: 0

             Total is: 3

     

     List = [_A,_B,_C] ?

The key to understanding what is going on is to understand what is happening to the first argument. In the query, the first argument is an uninstantiated variable: 

     | ?- demo_count_elems_acc(List, 0, 3).
Remember that an uninstantiated variable will unify with anything. Now look at the heads of the two rules: 

     % 1 terminating condition

     count_elems_acc([], Total, Total).

     % 2 recursive

     count_elems_acc([Head|Tail], Sum, Total) :- ...

We can test the unification of the heads of the rules with the query by typing the following into the Prolog interpreter: 

     | ?- count_elems_acc([], Total, Total) = count_elems_acc(List, 0, 3).

     no

     | ?- count_elems_acc([Head|Tail], Sum, Total) = count_elems_acc(List, 0, 3).

     Sum = 0,

     List = [Head|Tail],

     Total = 3 ? ;

     no

The first fails because Total can't simultaneously be instantiated to 0 and 3; the second succeeds because this condition isn't specified. The interesting thing is that List has been instantiated to [Head|Tail], ie a list with variables as head and tail. 

We could repeat this exercise for each stage of recursion, but next we'll look at the unification of the heads when the recursive call is: 

     count_elems_acc(_1481, 3, 3).

As before, we'll try each in turn: 

     | ?- count_elems_acc([], Total, Total) = count_elems_acc(_1481, 3, 3).

     Total = 3 ? ;

     no

     | ?- count_elems_acc([Head|Tail], Sum, Total) = count_elems_acc(_1481, 3, 3).

     Sum = 3 

     Total = 3 

     no

We would expect the terminating condition to unify, but we may be surprised that the recursive condition also unifies with the sub-goal. (It unifies because Sum and Total may or may not have the same instantiation.) 

We'll now look at what happens if we require a second solution to the query: 

     count_elems_acc(List, 0, 3).

The output of the demo version of the procedure restarts at depth 4, the level of recursion of the terminating condition for this query: 

                    Before recursion at depth 4

                       List is: [_2512|_2513]

                       Count is: 3

                       Total is: 3

                         Before recursion at depth 5

                            List is: [_3804|_3805]

                            Count is: 4

                            Total is: 3

                              Before recursion at depth 6

                                 List is: [_5356|_5357]

                                 Count is: 5

                                 Total is: 3

                                   Before recursion at depth 7

                                      List is: [_7168|_7169]

                                      Count is: 6

                                      Total is: 3

                                        Before recursion at depth 8

                                           List is: [_9240|_9241]

                                           Count is: 7

                                           Total is: 3

                                             Before recursion at depth 9

                                                List is: [_11572|_11573]

                                                Count is: 8

                                                Total is: 3

at which point, we are clearly in infinite recursion, so I stopped the run. 

Why has Prolog gone off into infinity? There are two kinds of answers to this. The first is at the practical level: when the second solution is requested, the recursive rule is used. This adds 1 to Count. Now Count is greater than Total and the terminating condition can never unify with anything this recursive rule will produce. 

The second kind of answer is the real point of this example. The procedure is poorly written. The terminating condition is fine: it specifies very precisely what is to happen when the counter and the total are equal. The recursive condition is poorly written. It clearly is supposed only to be applicable when the counter is less than the total, but this nowhere specified. To correct the problem, we should make a new version: 

     % 1 terminating condition

     count_elems_acc_correct([], Total, Total).

     % 2 recursive

     count_elems_acc_correct([Head|Tail], Sum, Total) :-

          Sum < Total,

          Count is Sum + 1, 

          count_elems_acc_correct(Tail, Count, Total).

When we try the query again, we get the solution we intend: 

     | ?- count_elems_acc_correct(List, 0, 3).

     List = [_A,_B,_C] ? ;

     no

	An alternative way of looking at clauses in a procedure
 

One way of thinking about clauses in a procedure is in terms of sets. Take count_elems_acc/3 as an example. Then we can let: 

        A = {all possible unifications with % 1}

        B = {all possible solutions with % 2}

So set A should include instances such as: 

     count_elems_acc([], 0, Total)

     count_elems_acc(List, 0, 3)

Set B should include 

     count_elems_acc([a,b,c], 0, Total)

     count_elems_acc(List, 0, 3)

From this rather inexhaustive example, we can see: 

In the rewritten version, count_elems_acc_correct/3, we achieved: 

If A union B is the universal set, then B = ¬A or to put it another way, B is the complement of A, written as A'. 

It is often useful to think about clauses within procedures in terms of set relations. However, it should be realised that one of Prolog's strengths is its capability to handle non-deterministic situations which, in turn, presupposes that clauses within procedures aren't mutually exclusive. 
  



	Multiple answers and memb/2

The definition of memb/3 given so far is: 

     % 1 terminating condition

     memb(Elem, [Elem|_]).

     % 2 recursive

     memb(Elem, [_|Tail]) :-

          memb(Elem, Tail).

We'll embed this within a small program which is rather trivial, but makes helps to make the relevant point: 

     small_prog(Elem, List, Output) :-

          memb(Elem, List),

          Output = Elem.

If we try a query such as: 

     | ?- small_prog(a, [a,b,b,a], Res).

     Res = a ? ;

     Res = a ? ;

     no

we can see that Prolog will give two solutions. It does this because the recursive clause will unify with anything that the terminating condition will unify with. This is illustrated by the following query which attempts to unify the heads of the two clauses of memb/3: 

     | ?- memb(Elem, [Elem|_]) = memb(Elem, [_|Tail]).

     true ? ;

     no

There are various views about this behaviour. It seems pointless to get into discussions about whether it is "correct" for Prolog to give one solution to examples like the above or whether it should give as many as possible: the answer presumably is that it depends on what you want the program to do. For some purposes it is merely necessary to know that a particular element occurs at least once in a list; for other purposes it is necessary to retrieve every instance. 

It should be remembered that when a program is backtracking, a non-deterministic procedure like memb/3 will be exhaustively searched each time it is encountered. This can be very time-consuming. 

It is possible to write deterministic version of memb/3 simply by insisting that within the recursive clause, it is required that the Elem being sought and the Head of the list don't unify: 

     % 1 terminating condition

     memb_det(Elem, [Elem|_]).

     % 2 recursive

     memb_det(Elem, [Head|Tail]) :-

          \+ (Elem = Head),

          memb_det(Elem, Tail).

'\+' is Prolog's way of writing "not". If we now use this procedure with small_prog/2, we can see that there is only one solution given. 

	Unification and equality again

The example of memb/2 and memb_det/2 both matched the sought Elem and the Head using unification (ie =/2). However, this is not the only criterion we may wish to use. For instance, we may want to insist on strict equality, ==/2. I've written both a non-deterministic and deterministic version: 

     % 1 terminating condition

     memb_eq(Elem, [Head|_]) :-

          Elem == Head.

     % recursive

     memb_eq(Elem, [_|Tail]) :-

          memb_eq(Elem, Tail).

     % 1 terminating condition

     memb_eq_det(Elem, [Head|_]) :-

          Elem == Head.

     % recursive

     memb_eq_det(Elem, [Head|Tail]) :-

          Elem \== Head,

          memb_eq_det(Elem, Tail).

We can illustrate the behaviour of these two procedures as follows: 

     | ?- Elem = a, memb_eq(Elem, [a,b,a,b]), Elem=Output.

     Elem = a 

     Output = a ? ;

     Elem = a 

     Output = a ? ;

     no

     | ?- Elem = a, memb_eq_det(Elem, [a,b,a,b]), Elem=Output.

     Elem = a 

     Output = a ? ;

     no

	More flexible procedures
 

It gets rather difficult to remember the names of the various procedures written to implement deterministic and non-deterministic membership with different kinds of matching. An alternative to having to remember the several names is to write a more versatile membership procedure that allows the matching to be specified in the arguments. 

We want to be able to write goals that look like: 

     | ?- memb_ext(a, [a,b,a,b], =).

     | ?- memb_ext(Var, [a,b,a,b], ==).

     | ?- memb_ext(3, [4,2,5,1,3], <).
and so on. 

One way to program this would be to write clauses for every type of matching, eg: 

     % 1 terminating - =/2 condition

     memb_ext(Elem, [Elem|_], =).

     % 2 terminating - ==/2 condition

     memb_ext(Elem, [Head|_], ==) :-

          Elem == Head.

     % 2 terminating - </2 condition

     memb_ext(Elem, [Head|_], <) :-

          Elem < Head.

etc.
This would be a time-consuming way of writing a program. A better way would be to, in some way, use the third argument in the running of the program. This is quite easy to do, providing we know a few more built-in predicates. 

The first thing to do is to check that the third argument of memb_ext/3 is instantiated. In fact, we can go further than this by insisting that it is instantiated to an atom and, if so, calling memb_ext/3: 

     % 1

     memb_ext(Elem, List, Relation) :-

          atom(Relation),

          memb_ext1(Elem, List, Relation).

memb_ext1/3 implements the membership relation. The simpler clause is the recursive one, which has the same pattern as the basic memb/3 procedure: 

     % 2 recursive

     memb_ext1(Elem, [_|List], Relation) :-

          memb_ext1(Elem, List, Relation).

The terminating condition is lengthier. Here we use three built-ins: functor/3, arg/3 and call/1: 

     % 1 terminating

     memb_ext1(Elem, [Head|_], Relation) :-

          functor(Relation1, Relation, 2),

          arg(1, Relation1, Elem),

          arg(2, Relation1, Head),

          call(Relation1).

The line functor(Relation1, Relation, 2) has the effect of forming whatever Relation is instantiated to into a structured object with two arguments. As an example, consider the following: 

     | ?- functor(Object, fred, 2).

     Object = fred(_A,_B) ? ;

     no

     | ?- functor(Obj, ==, 2).

     Obj = _A==_B ? ;

     no

The line arg(1, Relation1, Elem) has the effect of instantiating the first argument of Relation1 to Elem. Hence the following: 

     | ?- arg(1, _22768==_22769, a), write(_22768==_22769), nl.

     a==_104

     yes

The following line instantiates the second argument. The last line, call(Relation1) merely executes Relation1 as a Prolog goal. 

To illustrate the working of these procedures, consider the following: 

     | ?- Elem = a, memb_ext(a, [a,b,a,b], =), Elem = Output.

     Elem = a 

     Output = a ? ;

     Elem = a 

     Output = a ? ;

     no

     | ?- memb_ext(Var, [a,b,a,b], ==).

     no

     | ?- Elem = 3, memb_ext(3, [4,2,5,1,3], <), Elem = Output.

     Elem = 3 

     Output = 3 ? ;

     Elem = 3 

     Output = 3 ? ;

     no

Although the use of "templates" such as these is rather esoteric, the built-ins used in the terminating rule of memb_ext1/3 are extremely important in Prolog because they are widely used in building compilers for specialised languages in Prolog. 

	The pattern for this is: 

     % 1 terminating condition

     rule([]) :-

          some_processing([]).

     % 2 recursive

     rule([Head|Tail]) :-

          some_processing(Head),

          rule(Tail).

Now, this is only a pattern: the procedures we'll see in this section will all have this pattern as part of them, although there may be additions. 


	

	is_list/1

is_list/1 is one of the simplest of this family of procedures. The idea is that we can tell if an argument is a list if we can reduce it to an empty list: 

     % 1 terminating condition

     is_list([]) :-

          write('This is a list'),

          nl.

     % 2 recursive

     is_list([_|Tail]) :-

          % we don't have any processing here - which is why we can

          % use an anonymous variable (instead of Head) in the argument

          is_list(Tail).

Try the following goals: 

     | ?- is_list([]).

     | ?- is_list([a,b,c]).

     | ?- is_list([X,Y,Z]).

  


	count_elems/2

The obvious difference between the previous example and count_elems/2 is that the latter has two arguments. For those who have used other programming languages which include functions, this can be confusing. Prolog procedures aren't returned with a value setable by the programmer, as are functions. Values to be passed back have to be included in the argument list. count_elems/2 is a simple example of this. count_elems/2 has the straightforward purpose of counting the number of elements in a list. The procedure is slightly different from the basic pattern: 

     % 1 terminating condition

     count_elems([], Total) :-

          Total = 0.                      % processing

     % 2 recursive

     count_elems([Head|Tail], Count) :-

          count_elems(Tail, Sum),         % recursive

          Count is Sum + 1.               % processing

The main difference is that the order of the recursive and processing sub-goals in the recursive rule have been swapped. If we draw a tree to represent the processing of some goal, eg: count_elems([alpha, beta, gamma], Cnt). 

the most obvious difference is that it is a reflection of the typical tree shown earlier. This is simply because of the switch of the recursive and processing sub-goals in the second rule. 
  



	app/3

A common process in list processing is the need to run two lists together to make one list. This is usually known as appending or concatenating lists. So we might have the lists: 

     [alpha, beta, gamma]

and 

     [chi, psi, omega]

and require the final list: 

     [alpha, beta, gamma, chi, psi, omega]

Unfortunately it isn't possible to get this solution simply with the query: 

     | ?- List1 = [alpha, beta, gamma], List2 = [chi, psi, omega], 

          [List1|List2] = List3.
(If you don't know what the List3 will be instantiated to, then you should try the query for yourself.) 

Instead, we have to write a procedure that will add one list to another, element-by-element. We'll start the process by considering the terminating condition. The assumption implicit in this is that if the first of the original lists is empty, then the result list must be the same as the second original list: 

     % 1 terminating condition

     app([], List2, List3) :-

          List2 = List3.

More experienced Prolog programmers always attempt to do as much work as possible in the argument list, rather than writing extra sub-goals. In this example, we can unify List2 and List3 in the argument list simply by calling them the same name. So we can rewrite this rule as a fact: 

     % 1 terminating condition

     app([], List, List).

Before looking at the recursive clause, we'll look at the kinds of queries that this will unify with. Here is a sample: 

     | ?- app([], [chi, psi, omega], [chi, psi, omega]).

     yes

     | ?- app([], [chi, psi, omega], List3).

     List3 = [chi,psi,omega] ? ;

     no

     | ?- app(Var, [chi, psi, omega], List3).

     

     Var = [] 

     List3 = [chi,psi,omega] ? ;

     no

     | ?- app([], [], Result).

     Result = [] ? ;

     no

     | ?- app(Var, [], Result).

     Var = [] 

     Result = [] ? ;

     no

The first three are predictable. In the case of the third, the first argument of the query is an uninstantiated variable which is unified with [] in the program. The last two are more interesting. It would seem obvious that, if there is a need to write a definition of what is to be done is the first argument is an empty list, then there should be a definition of what is to be done if the second argument is an empty list. (I suppose we ought to also write a definition that would cope when both the first and second arguments are empty lists.) In practice, we don't need to do this as our definition will cover situations where the second list is either empty or not empty. 

This brings us to the recursive rule. We have just seen that we can deal with the second argument if it is empty or non-empty. This suggests that the recursive rule should work on the first argument, which we haven't yet dealt with. All we can really say about the first argument is that we want each element of it stuck onto the second argument. We've seen that we can't just "stick" the two lists together, so we are reduced to attaching the first list to the second, one element at a time. 

An initial definition that fits into our basic pattern is: 

     % 2 recursive

     app([Hd1|Tl1], List2, [Hd3|Tl3]) :-

          Hd1 = Hd3              % this makes the head of the first list

                                 % unify with the head of the third list

          app(Tl1, List2, Tl3).

We can read this as: appending a list is true if it is true that we can unify the head of the first list with the head of the third list and that we can append the tail of the first list, the second list and the tail of the third list. 

As with the definition of the terminating condition, we can rewrite this rule more succinctly, simply by renaming the variables used to stand for the head of the first and third lists: 

     % 2 recursive

     app([Hd|Tl1], List2, [Hd|Tl3]) :-

          app(Tl1, List2, Tl3).

If we put these two clauses together, we get a procedure that looks like: 

     % 1 terminating condition

     app([], List, List).

     % 2 recursive

     app([Hd|Tl1], List2, [Hd|Tl3]) :-

          app(Tl1, List2, Tl3).

Try out a few queries to ensure that this works: 

     | ?- app([alpha, beta, gamma], [chi, psi, omega], List3).

     | ?- app([], [chi, psi, omega], List3).

     | ?- app([alpha, beta, gamma], [], List3).

     | ?- app([], [], List3).
Instantiation of variables in app/3

We now have a working version of app/3. We'll now consider in detail how and when each argument in the app/3 procedure becomes instantiated. We could do this by using a trace of app/3, but instead we'll use a special version that includes some extra code. We'll call this demo_app/3. 

     | ?- demo_app([alpha, beta, gamma], [chi, psi, omega], List3).

          Before recursion at depth 1

             List1 is: [alpha,beta,gamma]

             List2 is: [chi,psi,omega]

             List3 is: [alpha|_4856]

               Before recursion at depth 2

                  List1 is: [beta,gamma]

                  List2 is: [chi,psi,omega]

                  List3 is: [beta|_5424]

                    Before recursion at depth 3

                       List1 is: [gamma]

                       List2 is: [chi,psi,omega]

                       List3 is: [gamma|_6243]

                         At the termination condition

                            List1 is: []

                            List2 is: [chi,psi,omega]

                            List3 is: [chi,psi,omega]

                    After recursion at depth 3

                       List1 is: [gamma]

                       List2 is: [chi,psi,omega]

                       List3 is: [gamma,chi,psi,omega]

               After recursion at depth 2

                  List1 is: [beta,gamma]

                  List2 is: [chi,psi,omega]

                  List3 is: [beta,gamma,chi,psi,omega]

          After recursion at depth 1

             List1 is: [alpha,beta,gamma]

             List2 is: [chi,psi,omega]

             List3 is: [alpha,beta,gamma,chi,psi,omega]

     

     List3 = [alpha,beta,gamma,chi,psi,omega] ;

     

     no

Look carefully at how List3 is instantiated. 

· At recursion of depth 1, the head is instantiated, but the tail isn't, so it is: [alpha|_4856]. 

· At depth 2, the head again is instantiated, but the tail isn't, so it is: [beta|_5424]. However, the alpha from the previous stage seems to have disappeared. This is because the recursive condition works on the tail of List3. 

· At depth 3, the same happens. Recursion is again on the tail of List3. 

· At the terminating condition, List3 is unified with List2, so is instantiated to the value: [chi,psi,omega]. 

· On the way "out" of the recursion, note that the tail of List3 is now instantiated. Also, note how the head is attached at each successive stage "out" of recursion.

	The pattern for this has two arguments, one being the element to be found, the other the list to be searched: 

     % 1 terminating condition

     rule(Elem, [Elem|_]) :-

          some_processing.

     % 2 recursive

     rule(Elem, [Head|Tail]) :-

          some_processing,          % it's unusual to have processing

                                    % in the recursive condition

          rule(Elem, Tail).

The major differences between this pattern and that for working through a list until it is empty lie in when any processing on the list is done. 

· in empty list termination, we are interested in every element in a list, eg to count them or to append them to another list. 

· in specified member termination, we are interested in the element that we find to terminate the computation: all other elements are usually irrelevant. 

As with empty list termination, this is only a pattern: the procedures we'll see in this section will all have this pattern as part of them, although there may be additions. 


	

	memb/2

Membership is the most widely given example in textbooks and it is the simplest. The idea is that we look through a list until we find that a specified item is the current head of the list. 

     % 1 terminating condition

     memb(Elem, [Elem|_]).

     % 2 recursive

     memb(Elem, [_|Tail]) :-

          % we don't have any processing here - which is why we can

          % use an anonymous variable (instead of Head) in the argument

          memb(Elem, Tail).

Try the following goals: 

     | ?- memb(gamma, [alpha, beta, gamma]).

     | ?- memb(alpha, [alpha]).

     | ?- memb(delta, [alpha, beta, gamma]).

     | ?- memb(alpha, []).
Inexperienced Prolog programmers often add an extra clause: 

     % 3 terminating condition (incorrect)

     memb(_, []).

If you add this to the previous procedure, you will find that the query: 

     | ?- memb(alpha, []).
will be successful. If this is what you want the procedure to do, then this is fine. However, most people feel that if an item doesn't exist in a list, then this should be failure. We don't need to do anything extra to program failure: Prolog will do this for us if we leave the condition out of the procedure. 
  


	memb/2 with a counter

memb/2 as given is often used and is frequently included in Prolog libraries. We can use it as the base of an example to show how to extend a procedure. Suppose we want to know the position at which an item occurs. We will accept that if the element occurs in the list of the goal, the procedure will write the position of that element (taking the first element as one, rather than zero), and that if an element doesn't occur in the list, then the goal will fail. 

We need to add some extra processing to both clauses and, of course, an extra argument. We'll look first at the recursive rule. Here we need just add one to a counter each time we recurse: 

     % 2 recursive

     memb(Elem, [_|Tail], Cnt) :-

          Cnt1 is Cnt + 1,

          memb(Elem, Tail, Cnt1).

The terminating clause can just write the position: 

     % 1 terminating condition

     memb(Elem, [Elem|_], Cnt) :-

          write(Cnt),

          nl.

Putting these two together, we have: 

     % 1 terminating condition

     memb(Elem, [Elem|_], Cnt) :-

          write(Cnt),

          nl.

     % 2 recursive

     memb(Elem, [_|Tail], Cnt) :-

          Cnt1 is Cnt + 1,

          memb(Elem, Tail, Cnt1).

Try the following goals: 

     | ?- memb(gamma, [alpha, beta, gamma], 1).

     | ?- memb(alpha, [alpha], 1).

     | ?- memb(delta, [alpha, beta, gamma], 1).

     | ?- memb(alpha, [], 1).
We have defined memb/2 and memb/3. Prolog doesn't confuse the two because, although they have the same functor, they have different arities. However, it is a bit irritating to have to type the ', 1' each time and we have to ensure that each time we get it correct. It is more convenient to write an extra, "calling" rule that supplies the extra argument for us. However, we can't call this memb/2, because this time Prolog will confuse it with our previous definition of memb/2, So we will call this one memb_count/2: 

     % 1

     memb_count(Elem, List) :-

          memb(Elem, List, 1).

Try the following goals to assure yourself they give the same solutions as before: 

     | ?- memb_count(gamma, [alpha, beta, gamma]).

     | ?- memb_count(alpha, [alpha]).

     | ?- memb_count(delta, [alpha, beta, gamma]).

     | ?- memb_count(alpha, []).

  


	delete_element/3

In the next example, an element is removed from a list. This is extremely similar to memb/2, but it is necessary to have a third argument which is used to represent the list without the specified element. 

     % 1 terminating condition

     delete_element(Elem, [Elem|Tail], Tail).

     % 2 recursive

     delete_element(Elem, [Head|Tail1], [Head|Tail2]) :-

          delete_element(Elem, Tail1, Tail2).

We'll look at the terminating condition first: 

     % 1 terminating condition

     delete_element(Elem, [Elem|Tail], Tail).

This is true when the specified element (Elem) is the head of the list being searched. When this is true, the third argument is the same as the tail of the list. 

The recursive rule looks more like app/3: 

     % 2 recursive

     delete_element(Elem, [Head|Tail1], [Head|Tail2]) :-

          delete_element(Elem, Tail1, Tail2).

In this rule, the Head of the list to be searched is unified with the Head of the list which represents the list minus the specified element. The recursive sub-goal specifies that Elem is in Tail1 with Tail2 being the list to be searched without Elem. 

Try the following goals to assure yourself they give the same solutions as before: 

     | ?- delete_element(gamma, [alpha, beta, gamma, delta], List).

     | ?- delete_element(alpha, [alpha], List).

     | ?- delete_element(delta, [alpha, beta, gamma], List).

     | ?- delete_element(alpha, [], List).
As with app/3, we can write a version of the procedure called demo_delete_element/3 that displays the instantiations of the arguments during recursion: 

     | ?- demo_delete_element(gamma, [alpha, beta, gamma, delta], List).

          Before recursion at depth 1

             Elem is: gamma

             "Search list" is: [alpha,beta,gamma,delta]

             "Result list" is: [alpha|_18743]

               Before recursion at depth 2

                  Elem is: gamma

                  "Search list" is: [beta,gamma,delta]

                  "Result list" is: [beta|_19292]

                    At the termination condition

                       Elem is: gamma

                       "Search list" is: [gamma,delta]

                       "Result list" is: [delta]

               After recursion at depth 2

                  Elem is: gamma

                  "Search list" is: [beta,gamma,delta]

                  "Result list" is: [beta,delta]

          After recursion at depth 1

             Elem is: gamma

             "Search list" is: [alpha,beta,gamma,delta]

             "Result list" is: [alpha,beta,delta]

     

     List = [alpha,beta,delta]

There are many variants on searching lists for a given element. Some of these are explored in the self-tests. 

	The pattern for this is: 

     % 1 terminating condition

     rule(1, Elem, [Elem|_]).

     % 2 recursive

     rule(Cnt, Elem, [_|Tail]) :-

          Cnt1 is Cnt - 1,

          rule(Cnt1, Elem, Tail).

Now, this is only a pattern: the procedures we'll see in this section will all have this pattern as part of them, although there may be additions. 


	

	position/3

position/3 is really just the basic pattern with no additions. The idea is that we know we have got to a particular position when the counter has been reduced to '1': 

     % 1 terminating condition

     position(1, Elem, [Elem|_]).

     % 2 recursive

     position(Cnt, Elem, [_|Tail]) :-

          Cnt1 is Cnt - 1,

          position(Cnt1, Elem, Tail).

Try the following goals: 

     | ?- position(0, Elem, [a,b,c]).

     | ?- position(1, Elem, [a,b,c]).

     | ?- position(2, Elem, [a,b,c]).

     | ?- position(3, Elem, [a,b,c]).

     | ?- position(4, Elem, [a,b,c]).

  


	delete_nth/3

This procedure is true if one list is the same as a second list, except that the nth element is deleted. As with count_elem/2, values to be passed back (such as the second list with the element deleted) have to be included in the argument list. 

     % 1 terminating condition

     delete_nth(1, [Elem|Tail], Tail).

     % 2 recursive

     delete_nth(Cnt, [Head|Tail1], [Head|Tail2]) :-

          Cnt1 is Cnt - 1,

          delete_nth(Cnt1, Tail1, Tail2).

Notice in the recursive clause how the Head of the first list is unified with the Head of the second list. You could use the Prolog tracer to follow how the variables are instantiated or add your own write/1 clauses to display variables during recursion. 

Try the following goals: 

     | ?- delete_nth(0,[a,b,c], List).

     | ?- delete_nth(1, [a,b,c], List).

     | ?- delete_nth(2, [a,b,c], List).

     | ?- delete_nth(3, [a,b,c], List).

     | ?- delete_nth(4, [a,b,c], List).

